
High energy spark 
This system drives the coil to obtain the maximum spark energy that is compatible with avoiding 

excessive erosion of the plugs. For starting, and during the first few revolutions when running, this rule is 

ignored and roughly twice the running energy is produced to ensure a healthy spark. The spark plug gap is 

set to 0.8 to 1.0 mm, standard for auto applications, twice which normally used for magnetos. 

 

 

Optimum advance timing 

The diagram above shows how the advance varies with RPM. This curve results in clean, smooth running 

at all speeds and a steady, reliable idle.  

Accurate timing 

As the system is driven direct from the crankshaft and as there are no mechanical drives or contact points, 

the timing point is free from small angular variations in timing (jitter). Thus the timing is accurate and 

remains free from drift with time.  

Idle trough 

As is with most modern ignition systems, the timing advances when the RPM falls below 600 RPM. The 

result is a steady and reliable idle. 

Easy starting, even with hand propping  
When starting, the spark plug fires at 3 degrees AFTER TDC. This makes kick-back impossible. 

Additionally, the spark energy is boosted. When the first few revolutions have been completed the 

advance is mapped to the curve above. This results in a very positive but gentle start.  

The spark energy is produced regardless of the speed at which the prop is pulled through TDC. You need 

only to pull the prop through TDC cleanly, not quickly.  

No adjustments, no moving parts 

It is a fit and forget system. There is nothing to wear. No adjustment is required or possible. It is tolerant 

of dirt. Maintenance consists of checking that the wiring connections and mountings are in good order. 

The battery capacity should be checked occasionally.  

Low power requirement.  
As it is energy efficient, each system requires only a maximum of 350 mA at 12V at 3,000 RPM.  

Compact and light  
As can be seen from the pictures and drawings. One controller weighs 100g, and one coil weighs 900g 

VAG (Bosch) coilpack are used as it is well tried, readily available and reasonably priced. Being vacuum 

impregnated with epoxy and having sealed EHT connector, they are very robust and water proof. They 

have a built in ignitor mounted on top. Individual ignition leads are available from accessory dealers.  

Each ignition controller drives one coil and one coilpack drives 4 plugs. No distributor is required. In 

order to be able to set the time for the next spark to occur, the ignition controller needs to know when the 

crankshaft passes the reference points. It gets this with the aid of a magnetic sensor and two magnets 180 

degrees apart that are fixed to a rotor on the crankshaft. One has the N pole facing out, the other has the S-

pole facing out. The N-pole corresponds to TDC for cylinders 1 and 3 (rear) and the S-pole corresponds 

to TDC for cylinders 2 and 4 (front).  

When the engine is running, the crankshaft rotates and the magnetic sensor produces two reference pulses 

each revolution. These are fed to a Arduino microcontroller, which records the time of each reference to 

milliseconds accuracy on an internal clock. From these times the RPM is calculated and the required 

timing advance angle is derived. The actual time of the next spark is calculated, and when the clock 

reaches this value, the appropriate pair of plugs fire. Which pair is fired depends on whether the last 

reference pulse was from a N-pole or a S-pole.  

For starting, the priority is to prevent kickback and ensure a big spark. To achieve this the spark occurs at 

3 degrees AFTER TDC. The active edge of each magnet is physically installed at this point and this 

generates the spark immediately.  

A spark occurs every time a piston approaches TDC. Thus one spark is “wasted” as it occurs at the end of 

the exhaust phase, and so has no effect. The benefit of this is that no distributor is required and the 

accuracy of the spark timing is improved, as the timing is taken directly from the crank, not through the 

gear driven camshaft.  

Tips and settings 
Set plug gaps to 0.9 to 1.0 mm  

Do not use batteries smaller than 7Ah for the main and 2.8 Ah for the backup. 

KEEP IT COOL  
The main enemy of reliability of electrical and electronic systems is heat. To avoid problems keep the 

temperature of the coils below 50 C when the engine is running, when air is being blown through the 

engine compartment. This may mean that special provision has to be made to direct fresh air at them. 

Compared to the cooling requirements of the engine, only a tiny amount is required. The Arduino 

controller is to be mounted behind the firewall inside the cabin. The Arduino controller needs to be inside 

a metal box and have all its wires shielded. 

Timing sensor 
The external sensor is connected to the controller with a shielded lead which has a connector at its other 

end. This plugs into the Arduino card. The external sensor is mounted within a 12.7 dia X 38 long (½” dia 

X 1 ½”long) AL tube. It can be fixed to the front of the VW crankcase with a suitable mount on the 

existing M6 tapped hole. 

The VCC of the sensor is connected to Arduino’s 5V power pin. The GND of the sensor is connected to 

the GND pin of Arduino. And the Vout or Signal pin of the hall effect sensor is connected to Arduino’s 

interrupt pin 0 (digital pin 2). Furthermore, a 10K resistor is connected between the VCC and Vout pins 

of the hall effect sensor. This is done to pull the output of the hall effect sensor to 5V.  

Coil pack 

 

Firing order 1-3-4-2  

Notes: Ignition coils and Power output stages are a combined unit. Ignition coil primary resistance 

cannot be measured. Secondary resistance can be measured (with ignition cables connected) by 

connecting a multimeter between the spark plug connectors of the ignition coil to be tested. Ignition 

cable RF suppression resistance is also measured. 

Ignition coils and ignition cables, checking. 

Disconnect 4 - pin harness connector from ignition coil Power output stage. - Disconnect spark plug 

connector. - Disconnect ignition cables from ignition coils. - Measure resistance of ignition wires and 

coils. 

Resistance of ignition wires Specified value: 4 to 6 k.  

Secondary resistance of ignition coils Specified value: 8 to 14 k. 

Secondary winding, checking - Check secondary winding on terminal 4 between cylinder 1 + cylinder 4, 

and cylinder 2 + cylinder 3. Specification: 4.0 to 6.0 k (at 20 C) If specification not obtained: - Replace 

ignition coils. 

 

Marked for ignition cables: A = Cylinder 1, B = Cylinder 2, C = Cylinder 3 and D = Cylinder 4 

Wasted spark theory 
While the transistors in the coil pack ground the coil's primary wire circuit (also called the Low Tension 

circuit), current and voltage increase through the primary windings and a magnetic field (flux) is 

produced around the coil and strengthens until the field maximizes. The time (traditionally measured in 

camshaft degrees or milliseconds) that the Arduino holds the coil's circuit grounded is called dwell 

(analogous to the points being closed in an older distributor system). In our coils, the (dwell) time 

required to maximize the magnetic field should be around 0.003 seconds (that is, 3 milliseconds or 3 ms). 

The ignition program 
The program is written in modified C++, some code is straight AVR processor specified code. A lot of 

effort has been made to make the code short and compact. The clock pulses of the AVR processor are 

used to measure the time. Since this system is a simple ignition system, the needs for microseconds are 

not present. So the ignition timing is measured in milliseconds. This published program has not yet been 

tested over a long period of time. It does function well as it is, but has not all the safety function built in 

yet. 

Calculations for the timing advance.  

RPM = Revolutions of crank per minute. 

V = Amount of degrees of crankshaft turn.  

t = Time in seconds for the crankshaft to turn. 

RPM = V / (6 x t) 

Example: How long time does it take for the crank to turn 10 degrees at 1000rpm? 

Given: V = 10, rpm = 1000: 

Formula: t = V / (6 x RPM) 

Answer= 10 / (6 x 1000) = 0,001667s = 1,667 ms 

RPM 

Timing advance 

BTDC 

Degrees from 

Hall sensor 

Time from Hall 

Sensor (ms) 

Timechange for 

each  RPM (ms) 

1 -3 0 0   

300 -3 180 100 0,334448161 

400 9 171 71,25 -0,2875 

600 8 172 47,77777778 -0,117361111 

1000 7 173 28,83333333 -0,047361111 

1700 17 163 15,98039216 -0,018361345 

2800 26 154 9,166666667 -0,006194296 

3100 28 152 8,172043011 -0,003315412 

4000 29 151 6,291666667 -0,002089307 

4100 0 180 7,317073171 0,010254065 

The program structure. 
The program consists of 5 sections:  

Declaration, Setup, Interrupt actions, Interrupt Service Routine and Loop.  

Declaration 

This is where all the variables are declared and deciding what data types they are and which pins to use. 

In this case we are manipulating the AVR processor built in Timers, so here is also where we include the 

AVR codes. 

Setup 

Here we setup all the variables in the terms of resetting, pointing out the pins and deciding if they are 

inputs or outputs. Her w also setup the interrupt functions to be used later in the program. The AVR timer 

no2 are set now and the correct type of interrupt service routine to be used with the timer are set. 

Interrupt actions 

This actions will happen when the hall sensors are exposed to the magnets on the crank. Interrupts are 

always interrupting the loop. There is one interrupt action for the rising trigger signal and on for the 

falling trigger signal. When magnet no 1 that has the north facing towards the sensor, flies by, the hall 

latches on and triggers the interrupt “SensorOn”. Deepening on the current rpm, the different things will 

happen. If the rpm is below 300, the Arduino will start the ignition system 2 straight away and start 

charging coil no 2 . If the rpm is over 300, ignition system 1 will start after the right amount of crank time 

has elapsed.  The timer will be initiated to be able to count up to 1 millisecond and automatically start 

over again. Also the RPM calculations will happen here. 

The second Interrupt “ SensorOff”, is like the first one, but opposite.  

Interrupt Service Routine 

This is when the timer has reached its counting compare match, in this case, one millisecond. For every 

time the timer has counted up to one millisecond, the variable “milliseconds” is added to one, and at the 

same time the program will check if something should happen. If the milliseconds is greater than the 

current crank time (which depends on the rpm calculations), the current coil will start to charge. After 3 

more milliseconds, the coil is discharged (Spark). After this, everyting is resetted, and the sequens is 

ready to start over. 

Loop. 

This is where the proper cranktime (Timing advance) is calculated. The Loop is executed whenever there 

are no interrupts active. 

/*****************Simple Electronic Ignition System by Anders Stenhammar. 2016***************************/ 

/************Declaration**************************************/ 

//Necessary to be able to manipulate the AVR processor timers: 

#include <Avr/interrupt.h> 

#include <Avr/io.h> 

 

//Pin definitions: 

const int HallPin1 = 2; // Pin nr 2 is conected to Hall sensor DO 

const int HallPin2 = 3; // Pin nr 3 is also conected to Hall sensor DO 

const int Ign1Pin = 8;  // (PORTB,0) 

const int Ign2Pin = 11; // (PORTB,3) 

volatile float milliseconds;         // The millisecondcounter value. (Volatile =Global variable) 

volatile float cranktime;             // Time from 3 degrees to where the coil should start to load. 

volatile float cranktimeStart;    // Variable to measure real cranktime. 

volatile float cranktimeStop;     // Variable to measure real cranktime. 

volatile float realcranktime;      // Measured real cranktime. 

volatile int IgnSystem;               // Statusword for active ign system. 

volatile float dwellTime;          // The time the coil should charge (default is 0,004 seconds. 

volatile unsigned long spark1;    // Counting the number of sparks given from coil 1 

volatile unsigned long spark2;    // Counting the number of sparks given from coil 2 

 

// RPM variables: 

volatile byte half_revolutions; 

volatile unsigned int rpm; 

volatile unsigned long timeold; 

 

/***********************************************************************************************/ 

void setup() { 

  Serial.begin(250000);        // Start the serial monitoring with the speed of 250000bpm 

  dwellTime = 3;               // Runningdwell: 3ms. Crankingdwell: 4ms. 

  half_revolutions = 0;         

  rpm = 0;                      

  timeold = 0;                 

  IgnSystem = 0;               // No ignition system is active. 

 

  pinMode(Ign1Pin, OUTPUT);    // Initialize the Ignition 1 pin as an output. 

  pinMode(Ign2Pin, OUTPUT);    // Initialize the Ignition 2 pin as an output. 

  bitClear(PORTB, 3);          // (digitalWrite(Ign2Pin, LOW);)  Turn the ignition off in case it's on 

  bitClear(PORTB, 0);          // (digitalWrite(Ign2Pin, LOW);)  Turn the ignition off in case it's 

 

  attachInterrupt(digitalPinToInterrupt(HallPin1), SensorOn, RISING);   //Hall sensor DO for Ignition 1 

  attachInterrupt(digitalPinToInterrupt(HallPin2), SensorOff, FALLING); //Hall sensor DO for Ignition 2 

 

  /*************** Setup timer2*****************************/ 

  noInterrupts(); 

  TCCR2A = 0;               // Turn off Control register for waveform generation 

  TCCR2B = 0;               // Turn off noise cancelling, turn off edge select, waveform gen mode 0, no clock source 

  TCCR2A |= (1 << WGM21);   // Turn on CTC mode (so it will start again) automatically 

  TIMSK2 |= (1 << OCIE2A);  // Set interrupt on compare match. 

  OCR2A = 249;              // ((0, 001 / 0.0000000625) / 64 ) - 1; //(0,001s=249) 

  TCNT2  = 0;               // Reset timer counter to 0 

  milliseconds = 0;         // Reset the ms counter variable. 

  interrupts(); 

} 

/*====================================================================================================== 

The interrupt action for magnet 1: The Timer starts to count up 1 ms at a time. 

*************************************************************************************************/ 

void SensorOn () { 

  if ((rpm < 300) & (IgnSystem == 0)) {   // When cranking (starting), the coil nr 2 will charge at once. 

    bitSet(PORTB, 3); //Turn on coil 1 charging immediately, Cranking dwell will then be 1ms longer than normal. 

    IgnSystem = 2;    // Statusword to know witch system is to be hot. 

  } 

  if ((rpm >= 300) & (IgnSystem == 2) ) {           // When running (not starting), the coil nr 1 will charge after cranktime in the ISR 

below. 

    IgnSystem = 1;                     //Statusword to know witch system is to be hot. 

  } 

  TCCR2B |= (1 << CS22);        // Load 64 prescaler / And this starts the timer2! 

  cranktimeStart = micros();  // For monitoring. 

 

  /********* RPM calculations: **********/ 

  half_revolutions++;        //For the RPM calculations 

  if (half_revolutions >= 2) { 

    rpm = 60 * 1000000 / (micros() - timeold) * half_revolutions; 

    timeold = micros(); 

    half_revolutions = 0; 

    realcranktime = (cranktimeStop - cranktimeStart); 

  } 

} 

/*====================================================================================================== 

 The interrupt action for magnet 2: The Timer starts to count up 1 ms at a time. 

*******************************************************************************************************/ 

void SensorOff () { 

  if ((rpm < 300) & (IgnSystem == 0))  {     // When cranking (starting), the coil nr 1 will charge at once. 

    bitSet(PORTB, 0);  //Turn on coil 2 charging immediately,  Cranking dwell will then be 1ms longer than normal. 

    IgnSystem = 1;     //Statusword to know witch system is to be hot. 

  } 

  if ((rpm >= 300) & (IgnSystem == 1) ) {            // When running (not starting), the coil nr 2 will charge after cranktime in the ISR 

below. 

    IgnSystem = 2;             //Statusword to know witch system is to be hot. 

  } 

  TCCR2B |= (1 << CS22);     // Load 64 prescaler / And this starts the timer2! 

  cranktimeStart = micros(); //For diagnostics 

  half_revolutions++;        //For the RPM calculations 

} 

/*====================================================================================================== 

 The Interrupt Service Routine for Timer2 that will be executed each time the timer reach the compare match register (1ms). 

*****************Milliseconds****************************************************************************/ 

ISR(TIMER2_COMPA_vect) { 

 

  /*********Counting milliseconds************************************************************/ 

  milliseconds++;  // Increases the variable "milliseconds" by 1 at each ms (every time the ISR is executed). 

 

  /************ coil charging*****************************************************************/ 

  if (milliseconds >= cranktime) {  //When the timer reaches the cranktime, then do this: 

 

    if (IgnSystem == 1) { //If ignitionsystem 1 is selected and not on, then: 

      bitSet(PORTB, 0);                                   //Turn on coil 1 charging. 

    } if (IgnSystem == 2) { //If ignitionsystem 2 is selected and not on, then: 

      bitSet(PORTB, 3);                                      //Turn on coil 2 charging. 

    } 

    cranktimeStop = micros();  //Stop recording the crank time 

  } 

  /***********Discharge coilspark*********************************************************************/ 

  //If the milliseconds has reached the cranktime and dwelltime, then: 

  if (milliseconds >= (cranktime + dwellTime)) { 

    milliseconds = 0;           // reset the ms counter variable 

    //Turn off timer. 

    TCCR2B  &= ~ (1 << CS22);   //clear the prescaler . 

    TCNT2 = 0;                  // Reset the timer count to 0 

 

    bitClear(PORTB, 0); //Stop charging coil 1. (Gives spark) 

    bitClear(PORTB, 3); //Stop charging coil 2. (Gives spark) 

 

    //Do one of this two alternatives for monitoring: 

    if (IgnSystem == 1) { //If ignitionsystem 1 is selected, then: 

      Serial.println ("Spark1"); // For monitoring 

    } 

    if (IgnSystem == 2) { //If ignitionsystem 2 is selected, then: 

      Serial.println ("Spark2");  // For monitoring 

    } 

    // If the engine has stopped or are still cranking, the IgnSystem is set to starting advance degrees. 

    if (rpm < 300) { 

      IgnSystem = 0; 

    } 

    Serial.println ("RPM"); 

    Serial.println (rpm);       // For monitoring 

  } 

} 

/******************************************************************************************************/ 

void loop() { 

 

 

  //Calculate the cranktime dependent on the rpm. (cranktime= advance time - dwellTime) 

  /********************Cranktime calculations ******************************************************* 

       RPM  Advance        Degrees             Cranktime     Time change 

            -degrees BTDC  from Sensor                        for each RPM (ms) 

       300    -3           180                 100              -100 

       400     9           171                 71,25            -0,2875 

       600     8           172                 47,77777778      -0,117361111 

       1000    7           173                 28,83333333      -0,047361111 

       1700    17          163                 15,98039216      -0,018361345 

       2800    26          154                 9,166666667      -0,006194296 

       3100    28          152                 8,172043011      -0,003315412 

       4000    29          151                 6,291666667      -0,002089307 

       4100    0           180                 7,317073171       0,010254065 

  */ 

  if (rpm <= 300) { 

    cranktime = 0; 

  } if ((rpm > 300) && (rpm <= 400)) { 

    cranktime = (100 - (rpm - 300) * 0.2875) - dwellTime; 

  } if ((rpm > 400) && (rpm <= 600)) { 

    cranktime = (71.25 - (rpm - 400) * 0.117361111) - dwellTime; 

  } if ((rpm > 600) && (rpm <= 1000)) { 

    cranktime = (47.77777778 - (rpm - 600) * 0.047361111) - dwellTime; 

  } if ((rpm > 1000) && (rpm <= 1700)) { 

    cranktime = (28.83333333 - (rpm - 1000) * 0.018361345) - dwellTime; 

  } if ((rpm > 1700) && (rpm <= 2800)) { 

    cranktime = (15.98039216 - (rpm - 1700) * 0.006194296) - dwellTime; 

  } if ((rpm > 2800) && (rpm <= 3100)) { 

    cranktime = (9.166666667 - (rpm - 2800) * 0.003315412) - dwellTime; 

  } if ((rpm > 3100) && (rpm <= 4000)) { 

    cranktime = (8.172043011 - (rpm - 3100) * 0.002089307) - dwellTime; 

  } 

  /****************************************/ 

} 

 


